

CONOSCERE IL COMPUTER DIRETTAMENTE DAL COMPUTER

per Commodore Vic20 e 64

CONOSCERE
IL COMPUTER
DIRETTAMENTE
DAL COMPUTER

per Commodore Vic20 e 64



Beatrice d'Este

CONOSCERE
IL COMPUTER
DIRETTAMENTE
DAL COMPUTER

per Commodore Vic20 e 64



Beatrice d'Este



Beatrice d'Este


L'argomento di questa lezione sono le tecniche avanzate di ingresso dati con le istruzioni **GET, READ, DATA, RESTORE**. Oltre all'INPUT (lez. 5) ci sono altre istruzioni che ti permettono di fornire dati al programma.

Quando viene eseguita un'istruzione INPUT il programma si ferma aspettando che dopo la risposta batti il RETURN.

Al contrario, usando l'istruzione GET, il programma continua l'esecuzione.

Il GET ti permette infatti di inserire un carattere per volta senza che debba premere il RETURN.


L'istruzione GET deve sempre essere seguita da una variabile (numerica o stringa), ad esempio GET A\$.



Durante l'esecuzione del GET A\$, il carattere battuto sulla tastiera, verrà assegnato ad A\$. Ma attenzione perchè il carattere battuto non comparirà sul video e per visualizzarlo dovrai stampare con un PRINT il contenuto nella variabile A\$. Se nessun carattere viene premuto, alla variabile dopo il GET verrà assegnato il numero o la stringa vuota (da non confondere con lo spazio), a seconda che sia una variabile numerica o stringa. Solitamente dopo il GET si usa una variabile stringa, infatti usandone una numerica quando viene premuto un tasto non numerico il computer segnala l'errore. Questo perchè, come ben sai, un carattere stringa non può essere assegnato in una variabile numerica. Per convertire il carattere inserito in un numero puoi usare la funzione VAL in questo modo:

GET A\$: A = VAL (A\$)

Il GET solitamente viene posto all'interno di un ciclo costruito con l'IF in modo che, se non vengono premuti tasti, ritorni ad eseguire il GET.



Osserva ora questo esempio. La linea:

10 GET A\$: IF A\$ = "" THEN 10

ha il compito di far attendere l'esecuzione del programma finchè non viene premuto un qualsiasi tasto.


Così, ad A\$ verrà assegnata la stringa vuota e il test

IF A\$ = "" THEN


... (cioè se A\$ è uguale alla stringa vuota) continuerà ad essere verificato.

A volte nei programmi capita di dover assegnare una grande quantità di dati, che solitamente rimangono costanti per tutta l'esecuzione. Potresti risolvere il problema scrivendo una lunga fila di assegnazioni, ma il procedimento non sarebbe molto efficace.

Quindi, quando devi assegnare dati a più variabili, a vettori o a matrici, ti conviene usare le istruzioni **READ, DATA e RESTORE**.



L'istruzione READ, che è seguita da una o più variabili numeriche o stringa, separate dalla virgola, legge nell'istruzione DATA i dati da assegnare alle variabili specificate. L'istruzione DATA, che è seguita da uno o più dati, numerici o stringa, separati dalla virgola, specifica i dati da assegnare alle variabili presenti nel READ. Così, se ci sono parecchi dati, dovrai usare più linee di programma con l'istruzione DATA, una in proseguimento dell'altra. Scrivi pure i DATA in qualsiasi punto del programma, visto che durante l'esecuzione verranno ignorati. Ogni volta che viene eseguito il READ, ad ogni variabile che segue verrà assegnato il dato successivo all'ultimo letto. Naturalmente ad ogni dato presente nel DATA deve corrispondere una variabile dello stesso tipo (numerica o stringa) nel READ. A volte può servire di dover rileggere i dati ripartendo dal primo: lo puoi fare scrivendo l'istruzione RESTORE.



Listato dell'esercizio: ESEMPIO PRATICO PER L'UTILIZZO DELLE ISTRUZIONI READ E DATA

```
10 DIM L(12)
20 FOR K=1 TO 12
30 READ L(K)
40 NEXT K
50 INPUT "MESE (1-12)"; M
60 IF M > 12 THEN 50
70 IF M = 0 THEN 50
80 INPUT "GIORNO"; G
90 IF G > L(M) THEN 80
100 FOR K=1 TO M: T=T+L(K): NEXT K: T=T-L(M)+G
110 PRINT "GIORNI TRASCORSI"; T
120 END
130 DATA 31,28,31,30,31,30,31,31,30,31,30,31
```

Listato dell'esercizio: COMPLETA IL LISTATO

```
10 |*| N$(12), L(12)
20 FOR K=1 TO |*|
30 READ N$(K), L(K)
40 NEXT K
50 INPUT "NUMERO MESE (1-12)"; |*|
60 IF M > 12 |*| 50
70 PRINT |*| " HA " L(M) " GIORNI "
80 PRINT "UN ALTRO MESE (S/N)"
90 GET |*|
100 IF A$ = |*| THEN 50
110 IF A$ = "N" THEN PRINT "FINE": |*|
120 GOTO |*|
130 DATA "GEN", 31, "FEB", |*|, "MAR", 31
140 DATA "APR", 30, " |*| ", 31, "GIU", 30
150 DATA "LUG", 31, "AGO", 31, "SET", 30
160 DATA "OTT", 31, "NOV", 30, "DIC", |*|
```

PROGRAMMIAMO INSIEME

```
10 PRINTCHR$(147)"PREMI UN TASTO"  
20 GETA$:IFA$=""THEN20  
30 TI$="000000"  
40 S$=RIGHT$(TI$,2)  
50 PRINTCHR$(147)"SECONDI: ";S$  
60 IFS$<>"15"THEN40  
70 PRINT"TEMPO FINITO"
```

Soluzione dell'esercizio "COMPLETA IL LISTATO" (apparso nella lezione n. 16)

```
10 PRINTCHR$(147)"SCRITTA GIREVOLE"  
20 INPUT"SCRITTA ";S$  
30 L=LEN(S$)  
40 IFL>15THEN10  
50 FORG=1TOL  
60 PRINTCHR$(147);S$;  
70 S$=RIGHT$(S$,L-1)+LEFT$(S$,1)  
80 FORK=1TO200:NEXTK  
90 NEXTG  
100 FORB=1TO3  
110 PRINTCHR$(147)"  
120 FORK=1TO200:NEXTK  
130 PRINTCHR$(147);S$  
140 FORK=1TO200:NEXTK  
150 NEXTB:GOTO50
```

COME UTILIZZARE LA GESTIONE MAGAZZINO

Con la GESTIONE MAGAZZINO puoi archiviare i prodotti (circa 300 sul CBM 64 e 25/30 sul VIC 20 senza espansione di memoria), fare l'inventario, calcolare il valore complessivo delle merci, ecc.

Come al solito, appena caricato, ti comparirà il MENU composto di dieci voci, che vedremo ora una per volta.

Per cominciare batti il n. 6 dell'INSERIMENTO. Ti verrà visualizzato il numero di CODICE dell'articolo che stai per inserire.

Inserisci quindi:	DESCRIZIONE	(usando al massimo 20 caratteri)
	PREZZO 1	(" " " 8 ")
	PREZZO 2	(" " " 8 ")
	ALIQUOTA IVA	(" " " 2 ")
	UNITA' DI MISURA	(" " " 6 ")
	GIACENZA	(" " " 8 ")

Il PREZZO 1 e il PREZZO 2 potrai usarli rispettivamente come: prezzo di listino e prezzo scontato. L'UNITA' DI MISURA può essere abbreviata, ad esempio: PZ-pezzi, NR-numero, LT-litri, MT-metri, ecc.

Durante l'INSERIMENTO ricorda di non lasciare mai in bianco le risposte, altrimenti anche battendo il RETURN non potrai proseguire.

Con il n. 7 puoi variare un articolo inserito e con il n. 8 eseguire la cancellazione.

Puoi ricercare un articolo introducendo, con il n. 1, il numero di CODICE. Se non lo ricordi, con il n. 2 la ricerca avverrà attraverso la DESCRIZIONE dell'oggetto inserito.

Per conoscere invece quali sono gli articoli che mancano o che stanno per andare in esaurimento nel magazzino, usa il n. 3 della SOTTOSCORTA e inserisci il numero del quantitativo che vuoi controllare.

Ad esempio, se premi il numero 20 ti verranno mostrati tutti gli articoli che hanno una giacenza uguale o minore a 20.

Con il n. 4 ti verrà mostrato l'INVENTARIO dell'intero magazzino e con il n. 5 verrà calcolato il valore totale delle merci contenute nel magazzino stesso.

Per registrare i dati inserisci il n. 10 della SCRITTURA FILE e scegli (battendo l'1 o il 2) il salvataggio sul nastro o sul floppy disk.

Una volta salvati i dati, premerai il n. 9 della LETTURA FILE se vuoi rileggerli.

Ricorda comunque che puoi fermare in qualsiasi momento il programma premendo RUN/STOP e RESTORE insieme.

Per tornare al programma senza perdere i dati inseriti scrivi GOTO 1000 e batti RETURN.